Offline EZI Example
This is a minimalist example of using EZI panels and tables in offline (or local) forms.  
It demonstrates  common header -> details type local form interactions. 
Requires RV16 on client and on RPG server. 
Local Tables Used
The HEADER Table

The ITEMS table

Form Views Used

RPG Programs Source Code
MJD_DOWN – Download sample HEADER and ITEMS rows
Defined as an online form view and attached to the schema’s main menu
     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
     H BNDDIR('EZISERVICE')
[bookmark: _GoBack]      /INCLUDE LRNGSOURCE,LRNGCOMMON
      /INCLUDE SOURCEDEMO,DEMOCOMMON
      /INCLUDE QRPGLESRC,EZISERVICE
      * ------------------------------
      * The HEADER table is defined as
      * ------------------------------
      *    HEAD01 - String (KEY)
      *    HEAD02 - String
      *    HEAD03 - String
      *    HEAD04 - String
      *    HEAD05 - String
      *    HEAD06 - String
      *    HEAD07 - String
      *    HEAD08 - String
      *    HEAD09 - String
      *    HEAD10 - String
      * -----------------------------
      * The ITEMS table is defined as
      * -----------------------------
      *    ITEM01 - String  (KEY)
      *    ITEM02 - Integer (KEY)
      *    ITEM03 - String
      *    ITEM04 - String
      *    ITEM05 - String
      *    ITEM06 - String
      *    ITEM07 - String
      *    ITEM08 - String
      *    ITEM09 - String
      *    ITEM10 - String
      * ----------------------------------------------------------
     D  MaxHeaders     C                   50
     D  MinItems       C                   3
     D  MaxItems       C                   30
      * ----------------------------------------------------------
     D  HeaderRows     S             10I00 Inz(0)
     D  ItemRows       S             10I00 Inz(0)
     D  Item           S             10I00 Inz(0)
     D  Rand           S             10I00 Inz(0)
      /INCLUDE SOURCEDEMO,STDPARAMS
      /free
        // ----------------------------------------------------------
        Exsr DownLoad_Data;
        EZI_OK_MessageBox('COMPLETION'
             : 'Uploaded ' + %char(HeaderRows) + ' HEADER table rows '
             + ' and ' + %char(ItemRows) + ' ITEMS table rows.' );
        LRNG_SendReceive(False:RequestPROGRAM:RequestACTION);
        *InLR = True;
        Return;
        // ----------------------------------------------------------
        BegSr DownLoad_Data;

          // Empty the client side tables - use 'Presend' SQL commands

          LRNG_UsingEx(False:'/DB.Presend');
            LRNG_SetProperty( 'SQL%1' : 'DELETE FROM HEADER');
            LRNG_SetProperty( 'SQL%2' : 'DELETE FROM ITEMS');
          LRNG_EndUsing();

           // Send the HEADER rows
           LRNG_SetProperty('/DB.Table%1' : 'HEADER');
           For HeaderRows = 1 to MaxHeaders;
             EZI_Offline_SendTableRow( 1
                                     : HeaderRows
                                     : 'HEAD01' : 'THING'+ %char(HeaderRows)
                                     : 'HEAD02' : EZI_RandomString('D')
                                     : 'HEAD03' : EZI_RandomString('D')
                                     : 'HEAD04' : EZI_RandomString('D')
                                     : 'HEAD05' : EZI_RandomString('D')
                                     : 'HEAD06' : EZI_RandomString('D')
                                     : 'HEAD07' : EZI_RandomString('D')
                                     : 'HEAD08' : EZI_RandomString('D')
                                     : 'HEAD09' : EZI_RandomString('D')
                                     : 'HEAD10' : EZI_RandomString('D') );
           EndFor;

           // Send ITEMS rows
           LRNG_SetProperty( '/DB.Table%2' : 'ITEMS');
           For HeaderRows = 1 to MaxHeaders;
              Rand = EZI_GetRandomNumber(MinItems:MaxItems);
              For Item = 1 to Rand;
                 ItemRows += 1;
                 EZI_Offline_SendTableRow( 2
                                         : ItemRows
                                         : 'ITEM01' : 'THING'+%char(HeaderRows)
                                         : 'ITEM02' : %char(Item)
                                         : 'ITEM03' : EZI_RandomString('D')
                                         : 'ITEM04' : EZI_RandomString('D')
                                         : 'ITEM05' : EZI_RandomString('D')
                                         : 'ITEM06' : EZI_RandomString('D')
                                         : 'ITEM07' : EZI_RandomString('D')
                                         : 'ITEM08' : EZI_RandomString('D')
                                         : 'ITEM09' : EZI_RandomString('D')
                                         : 'ITEM10' : EZI_RandomString('D') );
             EndFor;
           EndFor;
           HeaderRows -= 1;

           // Finished

        EndSr;
      /end-free
      
      
MJD_PANEL1 – HEADER row details
Defined as an offline/local form view, but not attached to the schema’s main menu

     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
     H BNDDIR('EZISERVICE')
      /INCLUDE LRNGSOURCE,LRNGCOMMON
      /INCLUDE QRPGLESRC,EZISERVICE
      /INCLUDE SOURCEDEMO,DEMOCOMMON
      * Define the Header details panel
     D Head_Panel      Ds                  LikeDS(EZI_QPanel) Inz(*LikeDS)
     D   Panel_Elem1   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem2   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem3   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem4   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem5   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem6   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem7   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem8   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem9   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem10  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)

      * Standard form view handler parameter list declaration
      /INCLUDE SOURCEDEMO,STDPARAMS

      /free
        // Bind everything at the form level
        LRNG_UsingEx(False:'/Form.&');
           LRNG_SetProperties('TBL: HEADER'
                             :'ASVIEW:'+ ProgramINFO.Name
                             :'SQL: SELECT ROWID, +
                                 HEAD01, HEAD02, HEAD03, HEAD04, HEAD05, +
                                 HEAD06, HEAD07, HEAD08, HEAD09, HEAD10  +
                                 FROM HEADER +
                                 WHERE ROWID = &(FLD:CURRENT_HEADER.ROWID)' );
        LRNG_EndUsing();
        // Define the overall form grid layout
        LRNG_SetProperties( '/Form.Grid.Col%1.Width:Fill'
                          : '/Form.Grid.Row%1.Height:Fill' );
        // Define the tables and panels on the form view
        Exsr Define_Header_Panel;
        // Send the form view template back and finish.
        LRNG_SendReceive(False:RequestPROGRAM:RequestACTION);
        *InLR = True;
        Return;

        // --------------------------------------------------------------------

        Begsr Define_Header_Panel;

          // Position on main form
          Head_Panel.ParentGRow = 1;
          Head_Panel.ParentGCol = 1;

          // Set up the panel elements
          Panel_Elem1.Label = 'Header Info 1';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem1 );

          Panel_Elem2.Label = 'Header Info 2';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem2 );

          Panel_Elem3.Label = 'Header Info 3';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem3 );

          Panel_Elem4.Label = 'Header Info 4';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem4 );

          Panel_Elem5.Label = 'Header Info 5';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem5 );

          Panel_Elem6.Label = 'Header Info 6';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem6 );

          Panel_Elem7.Label = 'Header Info 7';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem7 );

          Panel_Elem8.Label = 'Header Info 8';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem8 );

          Panel_Elem9.Label = 'Header Info 9';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem9 );

          Panel_Elem10.Label = 'Header Info 10';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem10 );

          // Define the panel
          Head_Panel.Title = 'Header Details Panel';
          EZI_QPanel_Define( Head_Panel );

          // Bind the panel elements to the SQL values retreived

          EZI_BindProperty( Panel_Elem1 : 'TEXT'  : 'FLD' : 'HEAD01');
          EZI_BindProperty( Panel_Elem2 : 'TEXT'  : 'FLD' : 'HEAD02');
          EZI_BindProperty( Panel_Elem3 : 'TEXT'  : 'FLD' : 'HEAD03');
          EZI_BindProperty( Panel_Elem4 : 'TEXT'  : 'FLD' : 'HEAD04');
          EZI_BindProperty( Panel_Elem5 : 'TEXT'  : 'FLD' : 'HEAD05');
          EZI_BindProperty( Panel_Elem6 : 'TEXT'  : 'FLD' : 'HEAD06');
          EZI_BindProperty( Panel_Elem7 : 'TEXT'  : 'FLD' : 'HEAD07');
          EZI_BindProperty( Panel_Elem8 : 'TEXT'  : 'FLD' : 'HEAD08');
          EZI_BindProperty( Panel_Elem9 : 'TEXT'  : 'FLD' : 'HEAD09');
          EZI_BindProperty( Panel_Elem10: 'TEXT'  : 'FLD' : 'HEAD10');

        Endsr;

      /end-free
      
MJD_PANEL2 – ITEMS  row details
Defined as an offline/local form view, but not attached to the schema’s main menu

     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
     H BNDDIR('EZISERVICE')
      /INCLUDE LRNGSOURCE,LRNGCOMMON
      /INCLUDE QRPGLESRC,EZISERVICE
      /INCLUDE SOURCEDEMO,DEMOCOMMON
      * Define the item details panel 1
     D Item_Panel1     Ds                  LikeDS(EZI_QPanel) Inz(*LikeDS)
     D   Panel_Elem1   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem2   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem3   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem4   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
      * Define the item details panel 2
     D Item_Panel2     Ds                  LikeDS(EZI_QPanel) Inz(*LikeDS)
     D   Panel_Elem5   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem6   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem7   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem8   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem9   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem10  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)

      * Standard form view handler parameter list declaration
      /INCLUDE SOURCEDEMO,STDPARAMS

      /free
        // Bind everything at the form level
        LRNG_UsingEx(False:'/Form.&');
           LRNG_SetProperties('TBL: ITEMS'
                             :'ASVIEW:'+ ProgramINFO.Name
                             :'SQL:SELECT ROWID, +
                               ITEM01,ITEM02,ITEM03,ITEM04,ITEM05,+
                               ITEM06,ITEM07,ITEM08,ITEM09,ITEM10 +
                               FROM ITEMS +
                               WHERE ROWID = &(FLD:CURRENT_ITEM.ROWID)' );
        LRNG_EndUsing();
        // Define the overall form grid layout
        LRNG_SetProperties( '/Form.Grid.Row%1.Height:Fill'
                          : '/Form.Grid.Row%2.Height:Fill' );
        // Define the two panels on the form view
        Exsr Define_Item_Panel1;
        Exsr Define_Item_Panel2;
        // Send the form view template back and finish.
        LRNG_SendReceive(False:RequestPROGRAM:RequestACTION);
        *InLR = True;
        Return;

        // --------------------------------------------------------------------

        Begsr Define_Item_Panel1;

          // Position on main form
          Item_Panel1.ParentGRow = 1;
          Item_Panel1.ParentGCol = 1;

          // Set up the panel elements
          Panel_Elem1.Label = 'Item Info 1';
          EZI_QPanel_DefineItem( Item_Panel1 : Panel_Elem1 );

          Panel_Elem2.Label = 'Item Info 2';
          EZI_QPanel_DefineItem( Item_Panel1 : Panel_Elem2 );

          Panel_Elem3.Label = 'Item Info 3';
          EZI_QPanel_DefineItem( Item_Panel1 : Panel_Elem3 );

          Panel_Elem4.Label = 'Item Info 4';
          EZI_QPanel_DefineItem( Item_Panel1 : Panel_Elem4 );

          // Define the panel
          Item_Panel1.Title = 'ITEM Details Panel 1';
          EZI_QPanel_Define( Item_Panel1 );

          // Bind the panel elements to the SQL values retreived
          EZI_BindProperty( Panel_Elem1 : 'TEXT'  : 'FLD' : 'ITEM01');
          EZI_BindProperty( Panel_Elem2 : 'TEXT'  : 'FLD' : 'ITEM02');
          EZI_BindProperty( Panel_Elem3 : 'TEXT'  : 'FLD' : 'ITEM03');
          EZI_BindProperty( Panel_Elem4 : 'TEXT'  : 'FLD' : 'ITEM04');

        Endsr;

        // --------------------------------------------------------------------

        Begsr Define_Item_Panel2;

          // Position on main form
          Item_Panel2.ParentGRow = 2;
          Item_Panel2.ParentGCol = 1;

          // Set up the panel elements
          Panel_Elem5.Label = 'Item Info 5';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem5 );

          Panel_Elem6.Label = 'Item Info 6';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem6 );

          Panel_Elem7.Label = 'Item Info 7';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem7 );

          Panel_Elem8.Label = 'Item Info 8';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem8 );

          Panel_Elem9.Label = 'Item Info 9';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem9 );

          Panel_Elem10.Label = 'Item Info 10';
          EZI_QPanel_DefineItem( Item_Panel2 : Panel_Elem10 );

          // Define the panel
          Item_Panel2.Title = 'ITEM Details Panel 2';
          EZI_QPanel_Define( Item_Panel2 );

          // Bind the panel elements to the SQL values retreived
          EZI_BindProperty( Panel_Elem5 : 'TEXT'  : 'FLD' : 'ITEM05');
          EZI_BindProperty( Panel_Elem6 : 'TEXT'  : 'FLD' : 'ITEM06');
          EZI_BindProperty( Panel_Elem7 : 'TEXT'  : 'FLD' : 'ITEM07');
          EZI_BindProperty( Panel_Elem8 : 'TEXT'  : 'FLD' : 'ITEM08');
          EZI_BindProperty( Panel_Elem9 : 'TEXT'  : 'FLD' : 'ITEM09');
          EZI_BindProperty( Panel_Elem10: 'TEXT'  : 'FLD' : 'ITEM10');

        Endsr;

      /end-free
    
MJD_STYLE1 – Browser HEADERs and associated ITEMs
Defined as an offline/local form view and attached to the schema’s main menu.

Defined as an offline/local form view
     H BNDDIR('LRNGSRV') DFTACTGRP(*no) ACTGRP(*caller)
     H BNDDIR('EZISERVICE')
      /INCLUDE LRNGSOURCE,LRNGCOMMON
      /INCLUDE QRPGLESRC,EZISERVICE
      /INCLUDE SOURCEDEMO,DEMOCOMMON
      * --------------------------------------------------------------------
      * Define the Head_Table
     D Head_Table      Ds                  LikeDS(EZI_QTable) Inz(*LikeDS)
     D   Head_Option1  Ds                  LikeDS(EZI_Image)   Inz(*LikeDS)     Note Image type
     D   Head_Column1  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Head_Column2  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
      * --------------------------------------------------------------------
      * Define the Header details panel
     D Head_Panel      Ds                  LikeDS(EZI_QPanel) Inz(*LikeDS)
     D   Panel_Elem1   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem2   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem3   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem4   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Panel_Elem5   Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
      * --------------------------------------------------------------------
      * Define the Item_Table
     D Item_Table      Ds                  LikeDS(EZI_QTable) Inz(*LikeDS)
     D   Item_Option1  Ds                  LikeDS(EZI_Image)   Inz(*LikeDS)     Note Image type
     D   Item_Column1  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Item_Column2  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Item_Column3  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
     D   Item_Column4  Ds                  LikeDS(EZI_Label)   Inz(*LikeDS)
      * --------------------------------------------------------------------
      * Standard form view handler parameter list declaration
      /INCLUDE SOURCEDEMO,STDPARAMS
      /free
        // Define the overall form grid layout
        LRNG_SetProperties( '/Form.Grid.Col%1.Width:Fill'
                          : '/Form.Grid.Col%2.Width:Fill'
                          : '/Form.Grid.Row%1.Height:Fill'
                          : '/Form.Grid.Row%2.Height:Fill' );
        // Define the tables and panels on the form view
        Exsr Define_Header_Table;
        Exsr Define_Header_Panel;
        Exsr Define_Items_Table;
        // Send the form view template back and finish.
        LRNG_SendReceive(False:RequestPROGRAM:RequestACTION);
        *InLR = True;
        Return;

        // --------------------------------------------------------------------
        Begsr Define_Header_Table;

          // Position on main form
          Head_Table.ParentGRow = 1;
          Head_Table.ParentGCol = 1;

          // Set up the table column details
          Head_Option1.Label = 'Det';
          Head_Option1.ResourceorIFSFile = 'tree_view_folder_open_24_gray.png';
          Head_Column1.Label = 'Head Col 1';
          Head_Column2.Label = 'Head Col 2';
          Head_Option1.Label = 'Details?';

          // Define the table
          Head_Table.Title = 'HEADER Summary Table - Touch for details';
          Head_Table.TitleAlign = 'Left';
          Head_Table.HighlightCurrentRow = True;
          Head_Table.FillColumn          = -1;

          EZI_QTable_Define( Head_Table
                           : Head_Option1
                           : Head_Column1
                           : Head_Column2 );

          // Make the HEADER material bind at the form level as CURRENT_HEADER
          LRNG_SetProperty('/Form.&.TBL'    : 'HEADER');
          LRNG_SetProperty('/Form.&.ASVIEW' : 'CURRENT_HEADER');
          LRNG_SetProperty('/Form.&.SQL'
                          : 'SELECT +
                             ROWID, HEAD01, HEAD02, HEAD03, HEAD04, HEAD05 +
                             FROM HEADER' );

          // Bind the table rowlist to the TOP_HEADER view of data
          EZI_BindProperty( Head_Table:'ROWLIST':'USINGVIEW':'CURRENT_HEADER');

          // Bind the table columns to the SQL columns
          EZI_BindProperty( Head_Column1 : 'TEXT'  : 'FLD' : 'HEAD01');
          EZI_BindProperty( Head_Column2 : 'TEXT'  : 'FLD' : 'HEAD02');

          // Handle touching a row in the header table
          LRNG_UsingEx(False:Head_Table.TableElement+'.OnRowClick.Operation');
             LRNG_SetProperties( 'Type:ExecSQL'
                               : 'RefreshForm:Y'
                               : 'SQL:?CURRENT_HEAD01 = SELECT &(FLD:HEAD01) ');
          LRNG_EndUsing();

          // Handle touching the open folder icon
          LRNG_UsingEx(False:Head_Option1.FullElementName+'.OnClick.Operation');
              LRNG_SetProperties( 'Type:OpenForm'
                                : 'Name:MJD_PANEL1'
                                : 'PresentMode:Stack');
          LRNG_EndUsing();

        Endsr;

        // --------------------------------------------------------------------
        Begsr Define_Header_Panel;

          // Position on main form
          Head_Panel.ParentGRow = 1;
          Head_Panel.ParentGCol = 2;

          // Set up the panel elements
          Panel_Elem1.Label = 'Head 1';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem1 );

          Panel_Elem2.Label = 'Head 2';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem2 );

          Panel_Elem3.Label = 'Head 3';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem3 );

          Panel_Elem4.Label = 'Head 4';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem4 );

          Panel_Elem5.Label = 'Head 5';
          EZI_QPanel_DefineItem( Head_Panel : Panel_Elem5 );

          // Define the panel
          Head_Panel.Title = 'Header Summary Panel';
          EZI_QPanel_Define( Head_Panel );

          // Bind the panel elements to the SQL values retreived

          EZI_BindProperty( Panel_Elem1 : 'TEXT'  : 'FLD' : 'HEAD01');
          EZI_BindProperty( Panel_Elem2 : 'TEXT'  : 'FLD' : 'HEAD02');
          EZI_BindProperty( Panel_Elem3 : 'TEXT'  : 'FLD' : 'HEAD03');
          EZI_BindProperty( Panel_Elem4 : 'TEXT'  : 'FLD' : 'HEAD04');
          EZI_BindProperty( Panel_Elem5 : 'TEXT'  : 'FLD' : 'HEAD05');

        Endsr;

        // --------------------------------------------------------------------
        Begsr Define_Items_Table;

          // Position on main form
          Item_Table.ParentGRow = 2;
          Item_Table.ParentGCol = 1;
          Item_Table.Cols       = -1;

          // Set up the table column details
          Item_Option1.Label = 'Det';
          Item_Option1.ResourceorIFSFile = 'tree_view_folder_open_24_gray.png';
          Item_Column1.Label = 'Item Info 1';
          Item_Column2.Label = 'Item Info 2';
          Item_Column3.Label = 'Item Info 3';
          Item_Column4.Label = 'Item Info 4';

          // Define the table
          Item_Table.Title = 'ITEMS Summary table - Touch for details';
          Item_Table.TitleAlign = 'Left';
          Item_Table.HighlightCurrentRow = True;
          EZI_QTable_Define( Item_Table
                           : Item_Option1
                           : Item_Column1
                           : Item_Column2
                           : Item_Column3
                           : Item_Column4 );

          // Bind the ROWLIST property of the item table to the SQL command used to fill the table
          EZI_BindProperty( Item_Table : 'ROWLIST':'TBL'   :'ITEMS');
          EZI_BindProperty( Item_Table : 'ROWLIST':'ASVIEW':'CURRENT_ITEM');
          EZI_BindProperty( Item_Table : 'ROWLIST': 'SQL'
                   : 'SELECT +
                     ROWID, ITEM01, ITEM02, ITEM03, ITEM04 +
                     FROM ITEMS WHERE ITEM01 = &(VAR:CURRENT_HEAD01)');

          // Bind the table columns to the SQL values
          EZI_BindProperty( Item_Column1 : 'TEXT'  : 'FLD' : 'ITEM01');
          EZI_BindProperty( Item_Column2 : 'TEXT'  : 'FLD' : 'ITEM02');
          EZI_BindProperty( Item_Column3 : 'TEXT'  : 'FLD' : 'ITEM03');
          EZI_BindProperty( Item_Column4 : 'TEXT'  : 'FLD' : 'ITEM04');

          // Handle touching a row in the items table
          LRNG_UsingEx(False:Item_Table.TableElement+'.OnRowClick.Operation');
             LRNG_SetProperties( 'Type:OpenForm'
                               : 'Name:MJD_PANEL2'
                               : 'PresentMode:Stack');
          LRNG_EndUsing();
          // Handle touching the open folder icon
          LRNG_UsingEx(False:Item_Option1.FullElementName+'.OnClick.Operation');
              LRNG_SetProperties( 'Type:OpenForm'
                                : 'Name:MJD_PANEL2'
                                : 'PresentMode:Stack');
          LRNG_EndUsing();

        Endsr;

      /end-free

